
Handle terminal events
    

Custom components can have local state. To make them becomes interactive, they need to react to terminal events. There are two ways to achieve that:

  
    
  
  Init configuration


In component's init callback, we can return a configuration with events_subscription: true to subscribe events for the component.
defmodule Example do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: nil, events_subscription: true}

  @impl true
  def handle_event(event, state, _attrs) do
    case event do
      %Orange.Terminal.KeyEvent{code: {:char, "q"}} ->
        Orange.stop()
        state

      %Orange.Terminal.KeyEvent{code: {:char, char}} ->
        IO.puts("Key pressed: #{char}")
        state

      _ ->
        state
    end
  end

  @impl true
  def render(state, _attrs, _update) do
    rect style: [width: 20, height: 20] do
      "Hello"
    end
  end
end

# Start the application. To quit, press 'q'.
Orange.start(Example)
Your component can handle events with the handle_event/3 callback and returns the updated state.

  
    
  
  Orange.subscribe/1


If you want to manually subscribe, use Orange.subscribe/1.
defmodule Example do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: nil}

  @impl true
  def after_mount(_state, _attrs, _update) do
    Orange.subscribe(:root)
  end

  @impl true
  def handle_event(event, state, _attrs) do
    case event do
      %Orange.Terminal.KeyEvent{code: {:char, "q"}} ->
        Orange.stop()
        state

      _ ->
        state
    end
  end

  @impl true
  def render(state, _attrs, _update) do
    rect style: [width: 20, height: 20] do
      "Hello"
    end
  end
end

# Start the application. To quit, press 'q'.
Orange.start({Example, id: :root})

  
    
  
  Remove events subscription


To unsubcribe for events, use Orange.unsubcribe/1.

  
    
  
  Component focus mode


There are certain cases that you only want one component to receive events and prevent other components. Look at the above example, an input component should receive all the events and prevent users from quitting the application when hitting "q". To do that, we can set a component as focused with Orange.focus/1. Only the focused component can receive terminal events. To remove the focus status, use Orange.unfocus/1.
Supported events
	Keyboard events: see Orange.Terminal.KeyEvent for more details
	Terminal resize event: see Orange.Terminal.ResizeEvent for more details




  

    
Orange 
    



      
Terminal UI framework for Elixir.
A framework empowering developers to build complex terminal UI applications. Orange provides a declarative way to build UI components and manage terminal events. Orange offers stateful components which can encapsulate local state and logics, allows you to split your application into discrete sub-components. For example, here is a panel displays a random numbers every 1 second:
defmodule Panel do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: 0, events_subscription: true}

  @impl true
  def after_mount(_state, _attrs, update) do
    spawn_link(fn -> random_number(update) end)
  end

  defp random_number(update) do
    number = :rand.uniform(100)
    update.(number)
    Process.sleep(1000)
    random_number(update)
  end

  @impl true
  def handle_event(event, state, _attrs) do
    case event do
      %Orange.Terminal.KeyEvent{code: {:char, "q"}} ->
        Orange.stop()
        state

      _ ->
        state
    end
  end

  @impl true
  def render(state, _attrs, _update) do
    rect style: [width: 25, height: 10, border: true] do
      rect do
        "Random number: #{state}"
      end
    end
  end
end

# Start the application. To quit, press 'q'.
Orange.start(Panel)
[image: rendered result]

      


      
        Summary


  
    Functions
  


    
      
        focus(component_id)

      


        Focus a component by component_id.



    


    
      
        start(element)

      


        Start the runtime and render the UI root.



    


    
      
        stop()

      


        Stop the runtime



    


    
      
        subscribe(component_id)

      


        Subscribe to the event manager.



    


    
      
        unfocus(component_id)

      


        Unfocus a component by component_id.



    


    
      
        unsubscribe(component_id)

      


        Unsubscribe to the event manager.



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    focus(component_id)



  


  

Focus a component by component_id.
There can be only one focused component at a time. If a component is focused, it will receive all the events and prevent other components from receiving them. This is useful when you want a component to intercept all terminal events, for example, an input.

  



  
    
      
      Link to this function
    
    start(element)



  


  

Start the runtime and render the UI root.
The root component must be a custom component.

  



  
    
      
      Link to this function
    
    stop()



  


  

Stop the runtime

  



  
    
      
      Link to this function
    
    subscribe(component_id)



  


  

Subscribe to the event manager.
Subscribed components will receive terminal events. To unsubscribe, use unsubscribe/1.

  
    
  
  Examples


defmodule Example do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: nil}

  @impl true
  def after_mount(_state, _attrs, _update) do
    Orange.subscribe(:root)
  end

  @impl true
  def render(_state, _attrs, _update) do
    rect do
      "Hello"
    end
  end
end

Orange.start({Example, id: :root})

  



  
    
      
      Link to this function
    
    unfocus(component_id)



  


  

Unfocus a component by component_id.

  



  
    
      
      Link to this function
    
    unsubscribe(component_id)



  


  

Unsubscribe to the event manager.

  


        

      



  

    
Orange.Macro 
    



      
Macros for creating Orange primitive components
Currently, Orange supports only one primitive component: rect (rectangle). Primitive components are
the building blocks of Orange applications. They can be used to construct more complex components.
For layout management, Orange leverages the taffy Rust crate.
It supports two layouts in the CSS specifications:
	flex: Flexible box layout
	grid: Grid layout system


  
    
  
  Examples


Macro provides an ergonomic way to create Orange components. For example, the following code:
rect style: [width: 12, border: true, flex_direction: :row] do
  rect style: [color: :red] do
    "Hello"
  end

  rect do
    "World"
  end
end
will produce the following struct:
%Orange.Rect{
  children: [
    %Orange.Rect{children: ["Hello"], attributes: [style: [color: :red]]},
    %Orange.Rect{children: ["World"], attributes: []}
  ],
  attributes: [style: [width: 12, border: true, flex_direction: :row]]
}

  
    
  
  Components children


Orange components can have children, provided by the do block. The syntax took inspiration from HTML:
rect do
  rect do
    "Hello"
  end

  rect do
    "World"
  end
end
The children can also be a list of components. This is useful when you want to render a collection of components. For example:
rect do
  [
    rect do
      "Hello"
    end,

    rect do
      "World"
    end
  ]
end
Normal Elixir expression works inside the children block. For example, here's how to conditionally render components:
rect do
  if :rand.uniform() > 0.5 do
    "Hello"
  else
    "World"
  end
end

  
    
  
  Styling



  
    
  
  Display


The :display property controls how children are laid out. Supported values are:
	:flex (default) - Flexible box layout
	:grid - Grid layout

Flex Layout
When display: :flex, the following properties control the layout:
	:flex_direction - Direction of the flex container. See MDN docs for more info. Supported values:
	:row (default) - Main axis is horizontal. Cross axis is vertical.

	:column - Main axis is vertical. Cross axis is horizontal.



	:flex_grow - How much the component grows relative to siblings when there is extra space. See MDN docs for more info.

	:flex_shrink - How much the component shrinks relative to siblings when space is limited. See MDN docs for more info.

	:justify_content - Alignment along the main axis. See MDN docs for more info. Supported values:
	:start (default) - Pack items at the start

	:end - Pack items at the end

	:center - Center items

	:space_between - Evenly space items with first/last at edges

	:space_around - Evenly space items with equal space around

	:space_evenly - Evenly space items with equal space between

	:stretch - Stretch items to fill container



	:align_items - Alignment along the cross axis. See MDN docs for more info. Supported values:
	:start (default) - Align items at the start

	:end - Align items at the end

	:center - Center items

	:space_between - Evenly space items with first/last at edges

	:space_around - Evenly space items with equal space around

	:space_evenly - Evenly space items with equal space between

	:stretch - Stretch items to fill container




Grid Layout
When display: :grid, the following properties control the layout:
	:grid_template_rows - defines the rows in the grid. Takes a list of track sizes. See MDN docs for more info.
	:grid_template_columns - defines the columns in the grid. Takes a list of track sizes. See MDN docs for more info.

Track sizes can be:
	An integer - fixed number of cells

	A percentage string - percentage of container size (e.g. "50%")

	{:fr, n} - takes up n fraction of remaining space

	:auto - sized based on content

	{:repeat, count, size} - repeats the size specification count times


Child items can be positioned in the grid using:
	:grid_row - specifies grid row placement. See MDN docs for more info.

	:grid_column - specifies grid column placement. See MDN docs for more info.


Grid placement values can be:
	An integer - places at specific grid line

	{:span, n} - spans n tracks

	:auto - automatic placement

	{start, end} - explicit start/end placement where start/end is a grid placement (e.g. {2, {:span, 3}} means
start at row/column index 2 and ends at index 5)
rect style: [
  display: :grid,
  grid_template_columns: [100, {:fr, 1}, "50%"],
  grid_template_rows: ["33%", {:repeat, 2, {:fr, 1}}]
] do
  rect style: [grid_column: {1, 3}] do
    "Header"
  end
  rect style: [grid_row: {:span, 2}] do
    "Sidebar"
  end
end



  
    
  
  Sizing


	:width - the width of the component

	:height - the height of the component


The values for width and height can be:
	An integer - the number of cells in the terminal. For example: style: [width: 10]

	A percentage string - the size is equal to the given percentage of the parent size. For example: style: [width: "50%"]



  
    
  
  Padding


Padding for the render box's inner content. The values for padding can be:
	One integer - padding for all sides. For example: style: [padding: 1] means [padding_top: 1, padding_bottom: 1, padding_left: 1, padding_right: 1]

	Two integers tuple - padding vertical and padding horizontal. For example: style: [padding: {1, 2}] means [padding_top: 1, padding_bottom: 1, padding_left: 2, padding_right: 2]

	Four integers tuple - padding top, right, bottom, left respectively. For example: style: [padding: {1, 2, 3, 4}] means [padding_top: 1, padding_bottom: 3, padding_left: 4, padding_right: 2]



  
    
  
  Margin


Margin creates space around the render box. The values for margin follow the same format as padding:
	One integer - margin for all sides. For example: style: [margin: 1] means [margin_top: 1, margin_bottom: 1, margin_left: 1, margin_right: 1]

	Two integers tuple - margin vertical and margin horizontal. For example: style: [margin: {1, 2}] means [margin_top: 1, margin_bottom: 1, margin_left: 2, margin_right: 2]

	Four integers tuple - margin top, right, bottom, left respectively. For example: style: [margin: {1, 2, 3, 4}] means [margin_top: 1, margin_bottom: 3, margin_left: 4, margin_right: 2]



  
    
  
  Border


	:border - whether to render a border around the rect. Defaults to false

	:border_top - whether to render a top border. Defaults to true if border is true

	:border_bottom - whether to render a bottom border. Defaults to true if border is true

	:border_left - whether to render a left border. Defaults to true if border is true

	:border_right - whether to render a right border. Defaults to true if border is true

	:border_color - the color of the border. See Color section for supported colors



  
    
  
  Text modifiers


An array of modifiers which adjust the text style. Supported modifiers are:
	:bold
	:dim
	:italic
	:underline
	:strikethrough


  
    
  
  Color


	:color - the color of the component text. The color value can be inherited from the parent component. If the color value is not specified, the component will inherit the color from the parent component

	:background_color - the color of the component background


The values for :color and :background_color is a single atom representing the color. Supported colors are:
	:white
	:black
	:grey
	:dark_grey
	:red
	:dark_red
	:green
	:dark_green
	:yellow
	:dark_yellow
	:blue
	:dark_blue
	:magenta
	:dark_magenta
	:cyan
	:dark_cyan


  
    
  
  Line wrap


By default, the text will wrap to the next line if it exceeds the width of the component. To disable this behavior, set the :line_wrap attribute to false

  
    
  
  Position


Rect elements support positioning. Supported values are:
	{:fixed, top, right, bottom, left} - the rect will be fixed at the given position. The values are offset to the respective edge of the screen. For example:
rect position: {:fixed, 1, 2, 1, 2} do
  "Fixed position"
end
will render:
[image: rendered result]



      


      
        Summary


  
    Functions
  


    
      
        rect(attrs \\ [], do_block)

      


        Generates a Orange.Rect struct



    





      


      
        Functions

        


    

  
    
      
      Link to this macro
    
    rect(attrs \\ [], do_block)



      (macro)

  


  

Generates a Orange.Rect struct

  
    
  
  Options


	:style - style attributes for the rect. Supported keys are:
	:width - see Sizing section

	:height - see Sizing section

	:border - see Border section

	:padding - see Padding section

	:color - see Color section

	:background_color - see Color section

	:display - see [Display][#module-display] section

	:flex_direction - available for :flex display. See Flex Layout section

	:justify_content - available for :flex display. See Flex Layout section

	:align_items - available for :flex display. See Flex Layout section

	:flex_grow - available for :flex display. See Flex Layout section

	:flex_shrink - available for :flex display. See Flex Layout section

	:grid_template_rows - available for :grid display. See Grid Layout section

	:grid_template_columns - available for :grid display. See Grid Layout section

	:grid_row - available for :grid display. See Grid Layout section

	:grid_column - available for :grid display. See Grid Layout section



	:title - the title of the rect. If specified, it implies border is true. The title can be a string or a map with supported keys are:
	:text - the title text. This field is required

	:color - the title color. This field is optional

	:text_modifiers - the title text modifiers. See Text modifiers section. This field is optional

	:offset - an integer specifies the title offset from the left edge. This field is optional and defaults to 0



	:scroll_x - the horizontal scroll offset

	:scroll_y - the vertical scroll offset

	:position - the position of the rect. See Position section



  
    
  
  Examples


iex> import Orange.Macro
iex> rect style: [width: 5, border: true, flex_direction: :row] do
...>   "foo"
...>
...>   rect do
...>     "bar"
...>   end
...> end
%Orange.Rect{
  children: [
    %Orange.Rect{children: ["foo"], attributes: []},
    %Orange.Rect{children: ["bar"], attributes: []}
  ],
  attributes: [style: [width: 5, border: true, flex_direction: :row]]
}

  


        

      



  

    
Orange.Component behaviour
    



      
A behaviour module for implementing custom components.
A custom component is a self-contained UI component that can be reused across the application. Custom components are composed of other custom or primitive components.
Custom components can have internal state and receive terminal events to update their state. They also have hooks, which will be trigger at specific points in the component lifecycle, for example, after_mount/1 and after_unmount/1

  
    
  
  Examples


In order to implement a custom component, you need to define a module that implements the Orange.Component behaviour. As an example, we will implement a Counter component. We can use up and down button to increase or decrease the counter value.
defmodule Counter do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: 0, events_subscription: true}

  @impl true
  def handle_event(event, state, _attrs) do
    case event do
      %Orange.Terminal.KeyEvent{code: :up} ->
        state + 1

      %Orange.Terminal.KeyEvent{code: :down} ->
        state - 1

      _ ->
        state
    end
  end

  @impl true
  def after_mount(state, _attrs, update) do
    # Set counter value to 10 after 5 seconds
    spawn(fn ->
      Process.sleep(5000)
      update.(10)
    end)
  end

  @impl true
  def render(state, attrs, _update) do
    rect style: [width: "100%", height: "1fr", border: attrs[:highlight]] do
      rect do
        "Counter: #{state}"
      end
    end
  end
end
This component can be used as children for other components like this:
defmodule App do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: nil, events_subscription: true}

  @impl true
  def handle_event(event, state, _attrs) do
    case event do
      %Orange.Terminal.KeyEvent{code: {:char, "q"}} ->
        Orange.stop()
        state

      _ ->
        state
    end
  end

  def render(_state, _attrs, _update) do
    rect style: [width: 20, height: 20, border: true] do
      # No attributes
      Counter

      # Or with attributes
      {Counter, id: :counter, highlight: true}
    end
  end
end
[image: Rendered result]

  
    
  
  Update callback


There are two ways to update the component state:
	By returning a new state from the handle_event/3 callback

	By using the update callback


The update callback will be passed to the render/3 and lifecycle hooks. The update callback can be called with a new state or a function that receives the current state and returns the new state.
def render(state, attrs, update_callback)
  update_callback.(%{state | counter: state.counter + 1})
  update_callback.(fn state -> %{state | counter: state.counter - 1} end)
end

        event()

      


    


    
      
        state()

      


    


    
      
        ui_element()

      


    


    
      
        update_callback()

      


    





  
    Callbacks
  


    
      
        after_mount(state, attributes, update_callback)

      


        Lifecycle hook that is called after the component is mounted (first time render to the terminal).



    


    
      
        after_unmount(state, attributes, update_callback)

      


        Lifecycle hook that is called after the component is unmounted (remove from the terminal).



    


    
      
        handle_event(event, state, attributes)

      


        Receives the event, current state, and attributes and returns the new state.



    


    
      
        init(attributes)

      


        Receives the attributes map from parent component and returns the component configuration.



    


    
      
        render(state, attributes, update_callback)

      


        Receives the current state, attributes and update callback and returns the rendered UI element.



    





      


      
        Types

        


  
    
      
      Link to this type
    
    event()



  


  

      

          @type event() :: Orange.Terminal.KeyEvent.t()


      



  



  
    
      
      Link to this type
    
    state()



  


  

      

          @type state() :: map()


      



  



  
    
      
      Link to this type
    
    ui_element()



  


  

      

          @type ui_element() :: Orange.Rect.t()


      



  



  
    
      
      Link to this type
    
    update_callback()



  


  

      

          @type update_callback() :: (state() -> state()) | state()


      



  


        

      

      
        Callbacks

        


  
    
      
      Link to this callback
    
    after_mount(state, attributes, update_callback)



      (optional)

  


  

      

          @callback after_mount(state(), attributes :: map(), update_callback()) :: any()


      


Lifecycle hook that is called after the component is mounted (first time render to the terminal).

  



  
    
      
      Link to this callback
    
    after_unmount(state, attributes, update_callback)



      (optional)

  


  

      

          @callback after_unmount(state(), attributes :: map(), update_callback()) :: any()


      


Lifecycle hook that is called after the component is unmounted (remove from the terminal).

  



  
    
      
      Link to this callback
    
    handle_event(event, state, attributes)



      (optional)

  


  

      

          @callback handle_event(event(), state(), attributes :: map()) :: state()


      


Receives the event, current state, and attributes and returns the new state.

  



  
    
      
      Link to this callback
    
    init(attributes)



  


  

      

          @callback init(attributes :: map()) :: %{state: state(), events_subscription: boolean()}


      


Receives the attributes map from parent component and returns the component configuration.
Configuration keys are:
	state - The initial state of the component

	events_subscription - whether the component should subscribe to terminal events



  



  
    
      
      Link to this callback
    
    render(state, attributes, update_callback)



  


  

      

          @callback render(state(), attributes :: map(), update_callback()) :: ui_element()


      


Receives the current state, attributes and update callback and returns the rendered UI element.
Info
This function must either return a single component or nil.


  


        

      



  

    
Orange.Component.Input 
    



      
An uncontrolled input component.

  
    
  
  Attributes


	:on_submit - A callback triggered when input is submitted. This attribute is required.

	:submit_key - The keyboard key that will trigger submission. This attribute is optional and defaults to :enter. See Orange.Terminal.KeyEvent for supported values.

	:on_exit - A callback triggered when input is exited. This attribute is optional.

	:exit_key - The keyboard key that will unfocus the input. This attribute is optional and defaults to :esc. See Orange.Terminal.KeyEvent for supported values.
Info
The :submit_key and :exit_key can not be :backspace as it is reserved for deleting characters.


	:auto_focus - Whether to focus automatically after mount. This attribute is optional. If true, the :id attribute is required and the input will unfocus after submission.

	:prefix - The input prefix string. This attribute is optional.

	:style - The component style. This attribute is optional.



  
    
  
  Examples


defmodule Example do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: %{search_value: ""}}

  @impl true
  def render(state, _attrs, update) do
    rect style: [width: 50, height: 20] do
      {
        Orange.Component.Input,
        id: :input,
        on_submit: fn value -> update.(%{state | search_value: value}) end,
        submit_key: {:char, "x"},
        prefix: "Enter search value:",
        auto_focus: true
      }

      rect do
        "Search value: #{state.search_value}"
      end
    end
  end
end
  [image: rendered result]

      





  

    
Orange.Component.Modal 
    



      
An modal/dialog component which renders an overlay in the middle of the viewport.

  
    
  
  Attributes


	:open - Whether the modal is open or not. This attribute is required.

	:offset_x - The offset from the left and right edge of the screen. This attribute is required.

	:offset_y - The offset from the top and bottom edge of the screen. This attribute is required.
Info
When the offset_x/y is too big for the terminal size, the modal will not be rendered.


	:children - A list of elements used as content of the modal. This attribute is optional.

	:title - The title of the modal. See Orange.Macro.rect/2 for supported values. This attribute is optional.

	:style - The modal style. See Orange.Macro.rect/2 for supported values. This attribute is optional.



  
    
  
  Examples


defmodule Example do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: %{search_value: ""}}

  @impl true
  def render(state, _attrs, update) do
    modal_content =
      rect do
        "foo"
        "bar"
      end

    rect do
      rect do
        "Displaying modal..."
      end

      {
        Orange.Component.Modal,
        offset_x: 8,
        offset_y: 4,
        children: [modal_content],
        open: true
      }
    end
  end
end
  [image: rendered result]

      





  

    
Orange.Component.VerticalScrollableRect 
    



      
A component that renders a vertical scrollable rectangle. When the container is overflown, a scrollbar is rendered.

  
    
  
  Attributes


	:height - The height of the container rect. This attribute is required.
	:content_height - The height of the inner content. This attribute is required.
	:scroll_offset - The vertical scroll offset. This attribute is required.
	:children - A list of elements rendered inside the container rect. This attribute is required.
	:title - The title of the scrolling box. This attribute is optional.


  
    
  
  Examples


defmodule Example do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: nil}

  @impl true
  def render(_state, attrs, _update) do
    total_items = 30

    rect style: [width: 20, height: 20, border: attrs[:highlight]] do
      {
        Orange.Component.VerticalScrollableRect,
        height: 20,
        content_height: total_items,
        scroll_offset: 5,
        children: for(i <- 1..total_items, do: "Item #{i}")
      }
    end
  end
end
  [image: rendered result]
  For more examples, see examples/scrollable_rect.exs.

Terminal keyboard events.
See key_code/0 below for supported keys.

        key_code()

      


    


    
      
        key_modifier()

      


    


    
      
        t()

      


    





      


      
        Types

        


  
    
      
      Link to this type
    
    key_code()



  


  

      

          @type key_code() ::
  :backspace
  | :enter
  | :left
  | :right
  | :up
  | :down
  | :home
  | :end
  | :page_up
  | :page_down
  | :tab
  | :back_tab
  | :delete
  | :insert
  | :f1
  | :f2
  | :f3
  | :f4
  | :f5
  | :f6
  | :f7
  | :f8
  | :f9
  | :f10
  | :f11
  | :f12
  | :null
  | :esc
  | :caps_lock
  | :scroll_lock
  | :num_lock
  | :print_screen
  | :pause
  | :menu
  | :keypad_begin
  | {:char, String.t()}


      



  



  
    
      
      Link to this type
    
    key_modifier()



  


  

      

          @type key_modifier() :: :shift | :ctrl | :alt | :super | :hyper


      



  



  
    
      
      Link to this type
    
    t()



  


  

      

          @type t() :: %Orange.Terminal.KeyEvent{code: key_code(), modifiers: [key_modifier()]}


      



  


        

      



  

    
Orange.Terminal.ResizeEvent 
    



      
Terminal resize event.

      


      
        Summary


  
    Types
  


    
      
        t()

      


    





      


      
        Types

        


  
    
      
      Link to this type
    
    t()



  


  

      

          @type t() :: %Orange.Terminal.ResizeEvent{
  height: non_neg_integer(),
  width: non_neg_integer()
}

